

「1 月～12 月までの各月の気温の平均値」などを求めるプログラムの「みほん」

- Fortran 90 プログラムの規則
  - `implicit none` は「お約束」として必ずつけておく。
  - 変数を宣言する文（「非実行文」の一種）は、最初の「実行文」が現れる前に書く。
  - 1 行の長さは最大で 132 文字 まで。  
行の最後にある「&」記号は、次の行へと継続する印。もしこの制限を超えそうになったら、行の最後に「&」記号を置いて次の行を起す。
- 「配列」とは
  - いくつもの「変数」をまとめて作った 1 つの「かたまり」
  - 配列の「要素」の参照のしかた（「番地」を指定する）
- `do` ループ (処理の繰り返し)
- `if` 文 (条件分岐)
- 「総和」のとり方、データの数え方
- 平均をとるための手順の「分解」

```
program enshu03b
  implicit none
  integer :: seireki,joutai
  real,dimension(12) :: kion
  real :: kion_nen_heikin,kion_nen_souwa
  real :: kion_nen_bunsan,kion_nen_hensa
  real :: kion2_nen_souwa,kion2_nen_heikin
  real,dimension(12) :: kion_tsuki_heikin,kion_tsuki_souwa
  integer :: i
  integer :: nensuu
  do i = 1,12
    kion_tsuki_souwa(i) = 0.0
  end do
  nensuu = 0
  do
    read(*,'(i4,12f5.1)',iostat=joutai)seireki,kion
    if (joutai /= 0) exit
    nensuu = nensuu + 1
    do i = 1,12
      kion_tsuki_souwa(i) = kion_tsuki_souwa(i) + kion(i)
    end do
    kion_nen_souwa = 0.0
    kion2_nen_souwa = 0.0
    do i = 1,12
      kion_nen_souwa = kion_nen_souwa+kion(i)
      kion2_nen_souwa = kion2_nen_souwa+kion(i)**2
    end do
    kion_nen_heikin = kion_nen_souwa/12.0
    kion2_nen_heikin = kion2_nen_souwa/12.0
    kion_nen_bunsan = kion2_nen_heikin-kion_nen_heikin**2
    kion_nen_hensa = sqrt(kion_nen_bunsan)
    write(*,'(i4,13f6.1,2f6.2)')seireki,kion,kion_nen_heikin &
      ,kion_nen_bunsan,kion_nen_hensa
  end do
  do i = 1,12
    kion_tsuki_heikin(i) = kion_tsuki_souwa(i)/real(nensuu)
  end do
  write(*,'(i4,12f6.2)')nensuu,kion_tsuki_heikin
end program enshu03b
```